Croustillant
sur lit de calcul formel

PARABOLE ET TRIANGLE ÉQUILATÉRAL



Soit $\mathscr{C}_f$ une parabole quelconque. FGH est un triangle équilatéral dont les côtés (ou les droites qui les portent) sont tangents à la parabole.

I est le centre du triangle FGH.

idée : sur le site de Mathematica

à faire

- Construire une figure dynamique : conjecturer le lieu de I, et celui de chacun des sommets.
- Démontrer les conjectures. On pourra se placer dans un repère tel que la fonction f est définie sur $\mathbb{R}$ par $f(x) = x^2$ (avec $a \neq 0$).

PARABOLE ET TRIANGLE ÉQUILATÉRAL

Idée de construction :

- Définir la fonction f à l'aide de trois curseurs
- A est un point de $\mathscr{C}_f$
- Tracer $(\mathrm{T_A})$ tangente en A à $\mathscr{C}_f$
- Construire $(\mathrm{R_A})$ image de $(\mathrm{T_A})$ par rotation de centre A et d'angle $\frac{\pi}{3}$
- Calculer l'abscisse de B, point de $\mathscr{C}_f$ dont la tangente $(\mathrm{T_B})$ à $\mathscr{C}_f$ est parallèle à $(\mathrm{R_A})$. Les pentes seront égales, donc $x_\mathrm{B} = \frac{m_{(\mathrm{R_A})}-b}{2a}$.
- Construire $(\mathrm{R_B})$ image de $(\mathrm{T_B})$ par rotation de centre B et d'angle $\frac{\pi}{3}$
- ...

Reprendre la même construction avec Xcas, mais la fonction est définie par $f(x) = x^2$. En effet on peut toujours construire un repère tel que l'équation de la parabole soit de cette forme... C'est tant mieux, car avec $f(x) = ax^2 + bx + c$, Xcas plante !!

```
/* FGH triangle équilatéral dont les côtés sont tangents à une parabole */
/* (1) lieu de I, centre de gravité du triangle */
/* (2) lieu des sommets du triangle  */
f(x):=x^2
Cf:=plotfunc(f(x),x=-10..10)
assume(xa=[-3.0,-10.0,10.0,0.0])
A:=point(xa,f(xa))
Ta:=tangente(Cf,A)
Ra:=rotation(A,pi/3,Ta)
xb:=simplifier((slope(Ra))/2)
B:=point(xb,f(xb))
Tb:=tangente(Cf,B)
Rb:=rotation(B,pi/3,Tb)
xc:=simplifier((slope(Rb))/2)
C:=point(xc,f(xc))
Tc:=tangente(Cf,C)
F:=single_inter(Ta,Tb)
G:=single_inter(Tb,Tc)
H:=single_inter(Tc,Ta)
I:=isobarycenter(F,G,H)

simplifier(coordinates(I))
```

on obtient les coordonnées de I :
$$\left(\frac{4x_a^3 - 3x_a}{12x_a^2 - 1}\,;-\frac{1}{4}\right)$$
son ordonnée est constante : I décrit la droite d'équation $y = -\dfrac{1}{4}$
C'est la directrice de la parabole.

on obtient les coordonnées de E, F et G en fonction de x_a ; celles de F sont : $\left(\dfrac{-4x_a}{12a^2x_a^2-1}, \dfrac{-4x_a^2+3}{48x_a^2-4}\right)$

Pour exprimer y_{F} en fonction de x_{F}, il faut d'abord exprimer x_a en fonction de x_{F}, puis remplacer x_a par son expression en fonction de x_{F} dans y_a.

On obtient au final deux expressions possibles pour y_{F} :

$$y_{\mathrm{F}} = \frac{-4\sqrt{3x_{\mathrm{F}}^2+1}-5}{12} \text{ et } y_{\mathrm{F}} = \frac{4\sqrt{3x_{\mathrm{F}}^2+1}-5}{12}$$

En faisant de même pour les deux autres sommets, on s'aperçoit qu'ils décrivent le même lieu que F !

```
simplifier(coordinates(F))
/* on cherche une relation entre les coordonnées de  xf et yf */
/* on va exprimer xa en fonction de xf, puis remplacer  */
/* xa (en fonction de xf) dans yf */
/* comme l'expression de xf est en xa^2, il y a deux possibilités */
[xf1,xf2]:=solve(xf=abscissa(F),xa)
simplifier(subst(ordinate(F),xa=xf1))
simplifier(subst(ordinate(F),xa=xf2))
[xh1,xh2]:=solve(xh=abscissa(H),xa)
simplifier(subst(ordinate(H),xa=xh1))
simplifier(subst(ordinate(H),xa=xh2))
[xg1,xg2]:=solve(xg=abscissa(G),xa)
normal(subst(ordinate(G),xa=xg1))
normal(subst(ordinate(G),xa=xg2))
/* donc dans tous les cas avec xs abscisse des sommets, */
/*  on a : ys = (-4*sqrt(3*xs^2+1)-5)/12 ou ys = (4*sqrt(3*xs^2+1)-5)/12 */
/* élevant au carré */
plotimplicit(-16*(3*x^2+1)+(12*y+5)^2=0,[x=-5..5,y=-5..5])
normal(-16*(3*x^2+1)+(12*y+5)^2)
```

⚠ pour vérifier les expressions des fonctions avec GeoGebra, il faut mettre les curseurs b et c à 0.

Commande

PARABOLE ET TRIANGLE ÉQUILATÉRAL

abscisse, assume, coordonnees, fenêtre de géométrie, inter_unique, isobarycentre, normal, ordonnee, pente, pi, plotfunc, plotimplicit, point, rotation, simplifier, solve, subst, tangente.